Computergestuetzte Mathematik zur Analysis

Lektion 9 (12. Dezember)

```
> restart: with(plots):
```

ebene parametrische Plots

```
> po := plot(sqrt(1-x^2),x = -1 .. 1,color = coral,thickness=2)
  ;
  pu := plot(-sqrt(1-x^2), x = -1 .. 1, color = coral,
  thickness = 2);
  display([po, pu]);
```

$$po := PLOT(\ldots)$$

$$pu := PLOT(\ldots)$$



```
> plot([[x, sqrt(1-(x+1)^2), x = -2 .. -1], [x, x^2, x = -1 ..
  0]],thickness=2);
```




```
> plot([sin(t), cos(t), t = 0 .. 2*Pi], color = red, thickness
  = 2);
```




```
>
> plot([cos(3*t), sin(7*t), t = 0 .. 2*Pi], color =
  "DarkGreen", thickness = 2); #Lissajous-Figur
```




```
> Lis1 := [cos(7*t), sin(5*t), t = 0 .. 2*Pi];
```

$$Lis1 := [\cos(7\,t), \sin(5\,t), t = 0..2\,\pi] \quad \textbf{(1.1)}$$

```
> Lis2 := [cos(3*t), sin(7*t), t = 0 .. 2*Pi];
```

$$Lis2 := [\cos(3\,t), \sin(7\,t), t = 0..2\,\pi] \quad \textbf{(1.2)}$$

```
> plot([Lis1, Lis2], color = [green, red], thickness = 2);
```




```
> kreis := [cos,sin,0..2*Pi];
```

$$kreis := [\cos, \sin, 0..2\pi] \tag{1.3}$$

```
> w := seq([cos(2*Pi*j/5), sin(2*Pi*j/5)], j = 0 .. 5);
```

$$w := [1, 0], \left[\cos\left(\frac{2}{5}\pi\right), \sin\left(\frac{2}{5}\pi\right)\right], \left[-\cos\left(\frac{1}{5}\pi\right), \sin\left(\frac{1}{5}\pi\right)\right], \left[-\cos\left(\frac{1}{5}\pi\right), -\sin\left(\frac{1}{5}\pi\right)\right], \left[\cos\left(\frac{2}{5}\pi\right), -\sin\left(\frac{2}{5}\pi\right)\right], [1, 0] \tag{1.4}$$

```
> Pentagram := [seq(w[k], k = 1 .. 6)];
```

$$Pentagram := \left[[1, 0], \left[\cos\left(\frac{2}{5}\pi\right), \sin\left(\frac{2}{5}\pi\right)\right], \left[-\cos\left(\frac{1}{5}\pi\right), \sin\left(\frac{1}{5}\pi\right)\right], \left[-\cos\left(\frac{1}{5}\pi\right), -\sin\left(\frac{1}{5}\pi\right)\right], \left[\cos\left(\frac{2}{5}\pi\right), -\sin\left(\frac{2}{5}\pi\right)\right], [1, 0]\right] \tag{1.5}$$

```
> plot([kreis, Pentagram], thickness = 2, scaling =
  constrained,axes='none');
```



```
> Stern := seq([[0, 0], w[k]], k = 1 .. 5);
```

$$Stern := [[0,0],[1,0]],\left[[0,0],\left[\cos\left(\frac{2}{5}\pi\right),\sin\left(\frac{2}{5}\pi\right)\right]\right],\left[[0,0],\left[-\cos\left(\frac{1}{5}\pi\right),\sin\left(\frac{1}{5}\pi\right)\right]\right],\left[[0,0],\left[-\cos\left(\frac{1}{5}\pi\right),-\sin\left(\frac{1}{5}\pi\right)\right]\right],\left[[0,0],\left[\cos\left(\frac{2}{5}\pi\right),-\sin\left(\frac{2}{5}\pi\right)\right]\right] \quad (1.6)$$

```
> plot([kreis, Pentagram, Stern], thickness = 2, scaling =
  constrained, axes = 'none');
```



```
> Pentagram2 := [seq(w[2*k mod 5+1], k = 1 .. 6)];
```

$$Pentagram2 := \left[\left[-\cos\left(\frac{1}{5}\pi\right), \sin\left(\frac{1}{5}\pi\right)\right], \left[\cos\left(\frac{2}{5}\pi\right), -\sin\left(\frac{2}{5}\pi\right)\right], \left[\cos\left(\frac{2}{5}\pi\right), \sin\left(\frac{2}{5}\pi\right)\right], \left[-\cos\left(\frac{1}{5}\pi\right), -\sin\left(\frac{1}{5}\pi\right)\right], [1, 0], \left[-\cos\left(\frac{1}{5}\pi\right), \sin\left(\frac{1}{5}\pi\right)\right]\right] \quad (1.7)$$

```
> plot([kreis,Pentagram2], thickness = 2, scaling =
  constrained,axes='none');
```



```
> f1:= x-> (-x)^(3/2)*((3/2)^(sqrt(-x))-floor((3/2)^(sqrt(-x)))
  );
```

$$f1 := x \to (-x)^{3/2}\left(\left(\frac{3}{2}\right)^{\sqrt{-x}} - \mathrm{floor}\left(\left(\frac{3}{2}\right)^{\sqrt{-x}}\right)\right) \qquad \textbf{(1.8)}$$

```
> plot(f1(x),x=-19.7..0);
```




```
> p2:=plot([f1(x), -f1(x)],x=-19.7..0);
```

$$p2 := PLOT(\ldots) \tag{1.9}$$

```
> display(p2);
```




```
> p3:=plot([f1(x),x+19.7,x=-19.7..0],thickness=4,color=green);
  # Aufrichten des Weihnachtsbaums
```

$$p3 := PLOT(\ldots) \qquad \textbf{(1.10)}$$

```
> display(p3);
```




```
> p4:=plot([-f1(x),x+19.7,x=-19.7..0],thickness=4,color=green);
```

$$p4 := PLOT(\ldots) \tag{1.11}$$

```
> p5:=plot([[-2,0],[-2,-3],[2,-3],[2,0]],thickness=4,color=
  brown):
> display(p3,p4,p5,axes=none);
```



Raumkurven / Wiederholung

```
> restart:
> with(plots):
> kurve := (2 - cos(t/6))*cos(t), (2 - cos(t/6))*sin(t), t/8;
```

$$kurve := \left(2 - \cos\left(\frac{1}{6}\,t\right)\right)\cos(t),\ \left(2 - \cos\left(\frac{1}{6}\,t\right)\right)\sin(t),\ \frac{1}{8}\,t \qquad \textbf{(2.1)}$$

Achtung: kurve ist eine Folge

```
> spacecurve([ kurve, t = -6*Pi .. 6*Pi], numpoints = 300,
  thickness = 5);
```



```
> kurve := [t, k, exp(-k^2*t^2), t = -2 .. 2];
```

$$kurve := [t, k, e^{-k^2 t^2}, t = -2..2] \tag{2.2}$$

```
> kurvenmenge := { seq(kurve, k = 1 .. 7) };
```

$$kurvenmenge := \{[t, 1, e^{-t^2}, t = -2..2], [t, 2, e^{-4t^2}, t = -2..2], [t, 3, e^{-9t^2}, t = -2..2], [t, 4, e^{-16t^2}, t = -2..2], [t, 5, e^{-25t^2}, t = -2..2], [t, 6, e^{-36t^2}, t = -2..2], [t, 7, e^{-49t^2}, t = -2..2]\} \tag{2.3}$$

```
> spacecurve(kurvenmenge, axes =  frame, thickness = 3);
```




```
> kurve := (5+cos(21*t))*cos(2*t), (5+cos(21*t))*sin(2*t), sin
  (21*t);
```

$$kurve := (5+\cos(21\,t))\cos(2\,t),\ (5+\cos(21\,t))\sin(2\,t),\ \sin(21\,t) \qquad \textbf{(2.4)}$$

```
> spacecurve([kurve, t = 0 .. 2*Pi], numpoints = 500, thickness
  = 3);
```



```
> j := 'j':
> n := 500;
```

$$n := 500 \tag{2.5}$$

```
> rgb_wert := evalf(sin(j*Pi/n)^2), 0, evalf(cos(j*Pi/n)^2);
```

$$rgb\_wert := \sin(0.006283185308\,j)^2,\ 0,\ \cos(0.006283185308\,j)^2 \tag{2.6}$$

```
> for j from 1 to n do;
>   p1 := subs(t = (j-1)*2*Pi/n, [kurve]);
>   p2 := subs(t = j*2*Pi/n, [kurve]);
>   pl[j] := spacecurve( [p1, p2], color = COLOR(RGB,
  rgb_wert), thickness = 3):
> od:
> display(convert(pl, set));
```



Flaechen im Raum

```
> restart:
> profil :=  cosh(t); #
```

$$profil := \cosh(t) \tag{3.1}$$

```
> plot(cosh(t),t=-1..1);
```




```
> flaeche := [ t, cos(s)*profil, sin(s)*profil];
```

$$flaeche := [t, \cos(s)\cosh(t), \sin(s)\cosh(t)] \tag{3.2}$$

```
> plot3d(flaeche, s = 0 .. 2*Pi, t = -1 .. 1);
```



```
> plot3d(flaeche, s = 0 .. 2*Pi, t = -1 .. 1,color="DarkGreen",
  style=patchnogrid,lightmodel=light4,glossiness=0.1,viewpoint=
  "circleleft",orientation=[30,45]);
```